MATH 2270-2
Symmetric matrices and quadrics
December 4, 2001

Quadric surfaces: #22 page 511 of Kolman

```
> restart:
> with(linalg):with(plots):#for computations and pictures
  with(student):#to do algebra computations like completing the
  square
```

```
> A:=matrix(3,3,[4,2,-2,2,4,-2,-2,-2,8]);
  B:=matrix(1,3,[6,-10,2]);
  C:=matrix(1,1,[-9/4]);
```

$$A := \begin{bmatrix} 4 & 2 & -2 \\ 2 & 4 & -2 \\ -2 & -2 & 8 \end{bmatrix}$$

$$B := \begin{bmatrix} 6 & -10 & 2 \end{bmatrix}$$

$$C := \begin{bmatrix} \frac{-9}{4} \end{bmatrix}$$

```
> fmat:=v->evalm(transpose(v)&*A&*v
                  + B&*v +C): #this gives a 1 by 1 matrix

  f:=(x,y,z)->simplify(fmat(matrix(3,1,[x,y,z]))[1,1]):
       #this extracts the entry, simplifies it, and defines f
  f(x,y,z);  #this is f,hopefully
```

$$4\,x^2+4\,x\,y-4\,x\,z+4\,y^2-4\,y\,z+8\,z^2+6\,x-10\,y+2\,z-\frac{9}{4}$$

```
> eigenvalues(A); #compute eigenvalues
```

$$2,\ 4,\ 10$$

Thus we have an ellipsoid, a point, or the empty set.

```
> data:=eigenvects(A);
```

$$data := [4, 1, \{[1, 1, 1]\}], [2, 1, \{[-1, 1, 0]\}], [10, 1, \{[1, 1, -2]\}]$$

We can read of the axes of our ellipsoid from the eigenvectors, and adjust so that S is a rotation matrix:

```
> w1:=1/sqrt(3)*matrix(3,1,[1,1,1]):
  w2:=1/sqrt(2)*matrix(3,1,[1,-1,0]):
  w3:=1/sqrt(6)*matrix(3,1,[1,1,-2]):
  S:=augment(w1,w2,w3);
  det(S);
```

$$S := \begin{bmatrix} \frac{1}{3}\sqrt{3} & \frac{1}{2}\sqrt{2} & \frac{1}{6}\sqrt{6} \\ \frac{1}{3}\sqrt{3} & -\frac{1}{2}\sqrt{2} & \frac{1}{6}\sqrt{6} \\ \frac{1}{3}\sqrt{3} & 0 & -\frac{1}{3}\sqrt{6} \end{bmatrix}$$

$$\frac{1}{6}\sqrt{3}\,\sqrt{2}\,\sqrt{6}$$

```
> evalm(transpose(S)&*A&*S); #just checking
```

$$\begin{bmatrix} 4 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 10 \end{bmatrix}$$

```
> g:=(u,v,w)->simplify(fmat(evalm(S&*matrix(3,1,[u,v,w])))[1,1]):
       #change of coordinates
  g(u,v,w);
```

$$4\,u^2 + 2\,v^2 + 10\,w^2 - \frac{2}{3}\sqrt{3}\,u + 8\sqrt{2}\,v - \frac{4}{3}\sqrt{3}\,\sqrt{2}\,w - \frac{9}{4}$$

```
> completesquare(%,u);
  completesquare(%,v);
  completesquare(%,w);
```

$$4\left(u - \frac{1}{12}\sqrt{3}\right)^2 - \frac{7}{3} + 2\,v^2 + 10\,w^2 + 8\sqrt{2}\,v - \frac{4}{3}\sqrt{3}\,\sqrt{2}\,w$$

$$2\,(v + 2\sqrt{2})^2 - \frac{55}{3} + 4\left(u - \frac{1}{12}\sqrt{3}\right)^2 + 10\,w^2 - \frac{4}{3}\sqrt{3}\,\sqrt{2}\,w$$

$$10\left(w - \frac{1}{15}\sqrt{3}\,\sqrt{2}\right)^2 - \frac{93}{5} + 2\,(v + 2\sqrt{2})^2 + 4\left(u - \frac{1}{12}\sqrt{3}\right)^2$$

So it really will be an ellipsoid.

```
> implicitplot3d(f(x,y,z)=0,x=-5..1,y=-1..5,z=-2..2,
  axes=boxed);#I adjusted the ranges to get a good picture
```



#28)

```
> A:=matrix(3,3,[-8,16,-2,16,-8,-2,-2,-2,10]);
  B:=matrix(1,3,[0,0,0]);
  C:=matrix(1,1,[-24]);
```

$$A := \begin{bmatrix} -8 & 16 & -2 \\ 16 & -8 & -2 \\ -2 & -2 & 10 \end{bmatrix}$$

$$B := [0 \quad 0 \quad 0]$$

$$C := [-24]$$

```
> fmat:=v->evalm(transpose(v)&*A&*v
                  + B&*v +C): #this gives a 1 by 1 matrix

  f:=(x,y,z)->simplify(fmat(matrix(3,1,[x,y,z]))[1,1]):
       #this extracts the entry, simplifies it, and defines f
  f(x,y,z);  #this is f,hopefully
```

$$-8\,x^2 + 32\,x\,y - 4\,x\,z - 8\,y^2 - 4\,y\,z + 10\,z^2 - 24$$

```
> eigenvects(A);
```

$$[6, 1, \{[1, 1, 1]\}], [12, 1, \{[1, 1, -2]\}], [-24, 1, \{[-1, 1, 0]\}]$$

A one or two sheeted hyperboloid. Note there are no linear terms here, so you can tell it’s going to be 1-sheeted! (why?)

```
> implicitplot3d(f(x,y,z),x=-3..3,y=-3..3,z=-3..3,
  axes=boxed);
```



3
2
1
0
–1
–2
–3
–2
x
0
2
–3
–2
–1
0
y
1
2
3

>